TEMPLATE = lib

CONFIG += core gui x11 link_pkgconfig

QT += xml

TARGET = ccfprojectfileparser

CHECKARCH = $$(ECLIPSEARCH)
linux*:isEmpty(CHECKARCH) {
    error("ECLIPSEARCH环境变量未设置，请设置为OSGI支持的架构，如x86、x86_64、aarch64、mipsel。");
}

linux* {
    DESTDIR = ../../ide.ccf.ccproject.linux.$$(ECLIPSEARCH)/lib
}
else:win32 {
    DESTDIR = ../../ide.ccf.ccproject.win32.x86/lib
}

DEFINES += QT_NO_KEYWORDS

PKGCONFIG += gtk+-2.0 glib-2.0

INCLUDEPATH += $$(JAVADIR)/include \
    $$(JAVADIR)/include/linux \
    $$(QTDIR)/src/gui/kernel \
    $$PWD/QtCreator \
    $$PWD
    
LIBS += -L/usr/lib \
    -lX11 \
    -lXt \
    -lgdk-x11-2.0 \
    -lgtk-x11-2.0

include(QtCreator/proparser.pri)|error("proparser.pri文件不存在！")


HEADERS += ccfprojectfileparser.h\
    scopelist.h \
    detailsview.h \
    editorview.h

SOURCES += ccfprojectfileparser.cpp\
    scopelist.cpp \
    detailsview.cpp \
    editorview.cpp
